Simulation of Power-Semiconductor Thermal Characteristics Using PSpice

Goce L. Arsov, Ph.D. and Ljupco P. Panovski, Ph.D.

Faculty of Electrical Engineering, St. Cyril and Methodius University, 91000 Skopje, P.O.Box 574, Republic of Macedonia

Abstract - A PSpice model for simulation of the thermal behaviour of power electronics devices is proposed. The model can be incorporated into any electronic device, providing that the dissipating power can be determined using PSpice simulation. The simulation results show very good dynamic and static behaviour of the proposed model.

Keywords - PSpice simulation, thermal behaviour, semiconductor, thermal equivalent circuit

1 INTRODUCTION

Temperature affects many of the semiconductor device characteristics. Its most noticeable effect is on the reverse saturation current of a pn junction. It is necessary to keep the junction temperature below the maximum junction temperature that a semiconductor device can withstand without failing due to thermal runway.

The heat generated within the material of a power semiconductor device must be dissipated into a heat sink or, for short periods, it can be stored in the heat capacity of the cell materials. As the materials through which the heat flows have both thermal resistance and capacitance, the thermal behaviour of the device is quite analogous to that of an electrical transmission line. As a consequence, the thermal impedance between the junctions and the heat sink is time-dependent. Because of the complex nature of heat generation in the semiconducting material, it is essential that simplifying assumptions be made so that the problem can be made tractable.

Knowing that heat capacitance is analogous to electrical capacitance, thermal resistance to electrical resistance, power (heat) to current, and temperature to voltage, the thermal circuit operation may be analyzed in terms of its electrical analogue. Thus, one can apply many of the same methods of analysis used in electrical RC network analysis. It means that any program capable for RC circuit simulation can be used to estimate the junction temperature of a semiconductor device.

2 THE MODEL

The equivalent circuits of power semiconductor devices can be found in the literature[1]. In our analysis we have used the simplified equivalent thermal circuit for an SCR as shown in Fig. 1 [1], [2]. The thermal resistance (R) and thermal capacitance (C) parameters of the model can be estimated from the standard semiconductor transient thermal impedance curve using recursive identification as shown in [2], or by parameter calculation obtained from module physical dimensions and manufacturer's data.



Fig. 1 Simplified equivalent thermal circuit for a semiconductor device

To examine the behaviour of the proposed model we have used the same values for the thermal resistances and capacitances as in [2]:

$R_1 = 0.006\ ^oC/W$, $C_1 = 0.033\ Ws/^oC$,

$R_2 = 0.111\ ^oC/W$, $C_2 = 0.148\ Ws/^oC$,

$R_3 = 0.122\ ^oC/W$, $C_3 = 1.180\ Ws/^oC$,

$R_4 = 0.166\ ^oC/W$, $C_4 = 9.500\ Ws/^oC$.

$R_5 = 0.011\ ^oC/W$,

Fig. 2 shows a typical input power sequence and output ΔT_{jc} response for the model presented in Fig. 1. There is no difference between the ΔT_{jc} curve of Fig. 2 and the one shown in [2] (Fig. 8).



Fig. 2 Simulated ΔT_{jc} response for the typical power input sequence for the model presented in Fig. 1

3 SIMULATION RESULTS

We have incorporated the thermal equivalent circuit to various power semiconductor devices and the performed simulations using evaluation version of PSpice have shown quite accurate results. The main problem was how to introduce the dissipated power into the equivalent thermal circuit of selected semiconductor device.

We present below the results of the transient thermal behaviour of an SCR used in a simple phase controlled rectifier. The heat input has been performed by polynomial voltage controlled current generator. A unity resistor has been used to obtain a voltage equivalent of the current through the semiconductor device.

The PSpice statements are given in Table 1 and the simulation results showing the waveforms of the load voltage and device junction temperature are shown in Fig. 3.

TABLE 1 PSpice statements for simulation of the thermal behaviour of SCR

```
*Simulation of the transient thermal behaviour of an SCR

VANP 1 0 SIN (0 311 50)
VGATE1A 2 3 PULSE ( 0 1 2.5M 1U 1U 2M 10M )
VS 35 0 AC 0
RL 3 36 1
RLL 36 35 1
XSCR 1 3 2 SCR

*Thermal equivalent circuit
GT1 0 51 POLY 2 3 36 1 3 0 0 0 0 1 0
RT1 51 52 .00610
CT1 52 0 .03300
RT2 52 53 .111
CT2 53 0 .148
RT3 53 54 .122
CT3 54 0 1.180
RT4 54 55 .166
CT4 55 0 9.5
RT5 55 0 .011

.PROBE
.OPTIONS RELTOL=30M ITL5=0 ITL4=20
.TRAN .5MS 1000M 0 .5M UIC
.END
```



Fig. 3 Simulation results of the transient thermal response of a simple SCR Phase controlled rectifier

4 CONCLUSION

A PSpice model for the simulation of the thermal behaviour of power electronics devices is proposed. The model can be incorporated into any electronic device, providing that the dissipating power can be determined using PSpice simulation. It can be used to estimate the temperature of a power electronics device which can then be incorporated in the .TEMP command for further simulations. The model could be also very useful in educational purposes. The simulation results show very good dynamic and static behaviour of the proposed model.

BIBLIOGRAPHY

1. F.E. Gentry et all. "*Semiconductor Controlled Rectifiers*", Prentice Hall 1964
2. G. L. Skibinski and W. A. Sethares, "Thermal Parameter Estimation Using Recursive Identification", *IEEE Trans. Power. Electron*, Vol. 6, No 2, pp 228-239, 1991.